
 1

 
Geometric data: 
Internal blast in a rigid tank. The geometry is axisymmetric. The explosive charge is a 
cylinder of solid TNT. 
Boundary conditions: 
Tank walls are treated as rigid. FSR (fluid-structure rigid) conditions (see Part 3) are 
adopted along the outer surface of the fluid domain. 
Materials 
The explosive charge is modelled by the JWL (Jones-Wilkins-Lee) law. Detonation is 
initiated at the upper central point of the charge and is assumed to proceed at a speed 
of 6930 m/s. The rest of the tank is filled by air at atmospheric pressure. 
We want to study the effects of the explosion up to 1.8 ms of physical time. 
 
2D axisymmetric simulation: 
 
JWLS2G 
The mesh generation file is: 
 
*%siz 50 
* 
TITRE 'Explosion de TNT dans un caisson' ; 
* 
option echo 0 ; 
* 
OPTION DIME 2 ELEM QUA4 ; 
* 
OPTION SORTIE 'jwls2g.msh'; 
option trac PSC FTRA 'jwls2g_mail.ps' ; 
* 
r_cais =  965.00e-3 ; 
h_cais = 1560.00e-3 ; 
r_tnt  =   35.50e-3 ; 
h_tnt  =  142.00e-3 ; 
* 
nh1 = 10 ; 
nv1 = 50 ; 
nv3 = nv1 ; 
* 
* pour avoir des éléments carrés dans le tnt : 
nv2 = ENTIER (nh1 * (h_tnt/r_tnt)) ; 
nh = nv1+nv2+nv1 ; 
* idem pour l'air (mailles carrées dans l'angle) : 
nh2 = ENTIER (nh * r_cais / h_cais) ; 
nh2 = nh2 - nh1 ; 
* 
h_bas = 0.5*(h_cais - h_tnt) ; 
h_hau = h_bas + h_tnt ; 
r_mil = r_cais * nh1 /(nh1+nh2) ; 
h_mil = h_cais * nv1 / nh ; 
* 
OPTION DENS (r_tnt / nh1) ; 
p4 =   0.  h_bas ; 
p5 = r_tnt h_bas ; 

p7 =   0.  h_hau ; 
p8 = r_tnt h_hau ; 
* 
OPTION DENS (h_cais / nh) ; 
p1 =   0.     0. ; 
p2 = r_mil    0. ; 
p3 = r_cais   0. ; 
p6 = r_cais h_mil  ; 
p9 = r_cais (h_cais - h_mil) ; 
p10=   0.   h_cais ; 
p11= r_mil  h_cais ; 
p12= r_cais h_cais ; 
* 
l_45 = p4 DROIT nh1 p5 ; 
l_58 = p5 DROIT nv2 p8 ; 
l_87 = p8 DROIT nh1 p7 ; 
l_74 = p7 DROIT nv2 p4 ; 
* 
l_12 = p1 DROIT nh1 p2 ; 
l_23 = p2 DROIT nh2 p3 ; 
l_36 = p3 DROIT nv1 p6 ; 
l_69 = p6 DROIT nv2 p9 ; 
l_912= p9 DROIT nv3 p12; 
l_1211= p12 DROIT nh2 p11 ; 
l_1110= p11 DROIT nh1 p10 ; 
* 
nv1 = 0 - nv1 ; 
nh2 = 0 - nh2 ; 
nv3 = 0 - nv3 ; 
* 
l_41 = p4 DROIT nv1 p1 ; 
l_52 = p5 DROIT nv1 p2 ; 
l_56 = p5 DROIT nh2 p6 ; 
l_89 = p8 DROIT nh2 p9 ; 
l_811= p8 DROIT nv3 p11; 

l_710= p7 DROIT nv3 p10; 
* 
l_axe = l_41 ET L_74 ET L_710 ; 
l_bas = l_12 ET l_23 ; 
l_hau = l_1211 ET l_1110 ; 
l_bor = l_36 ET l_69 ET l_912 ; 
l_cont = l_axe ET l_bas ET l_bor ET l_hau ; 
* 
tnt = l_45 l_58 l_87 l_74 DALLER PLAN ; 
tnt = tnt COUL ROUGE ; 
* 
air1 = l_12 l_52 l_45 l_41 DALLER PLAN ; 
air2 = l_23 l_36 l_56 l_52 DALLER PLAN ; 
air3 = l_56 l_69 l_89 l_58 DALLER PLAN ; 
air4 = l_89 l_912 l_1211 l_811 DALLER PLAN ; 
air5 = l_87 l_811 l_1110 l_710 DALLER PLAN ; 
* 
air = air1 ET air2 ET air3 ET air4 ET air5 ; 
air = air COUL TURQUOISE ; 
* 
*-- capteurs : 
e_p1  = air ELEM CONTENANT p1 ; 
e_p3  = air ELEM CONTENANT p3 ; 
e_p4  = air ELEM CONTENANT p4 ; 
e_p7  = air ELEM CONTENANT p7 ; 
e_p10 = air ELEM CONTENANT p10 ; 
e_p12 = air ELEM CONTENANT p12 ; 
e_bas  = air ELEM CONTENANT (BARY (p1 ET p3)) ; 
e_face = air ELEM CONTENANT (BARY (p3 ET p12)) ; 
e_haut = air ELEM CONTENANT (BARY (p10 ET p12)) ; 

e_mil  = air ELEM CONTENANT (BARY (air ET tnt)) ; 
e_sup  = air ELEM CONTENANT (BARY (p7 ET p10)) ; 
e_inf  = air ELEM CONTENANT (BARY (p1 ET p4)) ; 
e_capt = e_p1 ET e_p3 ET e_p4 ET e_p7 ET e_p10 ET e_p12 ET 
         e_bas ET e_face ET e_haut ET e_mil ET e_sup ET e_inf ; 
* 
tout = air ET tnt ; 
sort tout ; 
* 
trac tnt ; 
trac tout ; 
trac QUALIF tout ; 
trac qualif e_capt ; 
trac qualif l_cont ; 
opti trac mif ftra 'jwls2g_mail.mif'; 
trac tnt ; 
trac tout ; 
trac QUALIF tout ; 
trac qualif e_capt ; 
trac qualif l_cont ; 
* 
mess 'NB_POIN        =' (NBNO tout) ; 
mess 'NB_CAR1 (tnt)  =' (NBEL tnt) ;  
mess 'NB_CAR1 (total)=' (NBEL tout) ;  
* 
**option donn 5 ; 
* 
option echo 1 ; 
* 
fin ; 

 
The input file is: 
 
Explosion d'une charge de TNT dans un caisson rigide ( JWLS ) 
* 
ECHO 
* 
GIBI TOUT 
* 
AXIS NONL ALE  
* 
DIMENSION 
    PT2L 12300  zone    2 
    FL24 12100 
    LIAIS 910  FSA   230 
    NALE    1  NBLE 12300    ndvc 45620 
    MTTI 30 
TERM 
* 
GEOMETRIE 
   FL24 tnt 
   FL24 air 
TERM 
* 
GRILLE  
     EULER LECT tnt TERM 
     ALE   TOUS 
     MEAN  AUTRES       
* 
MATERIAU 
**   l'air :  on calcule eint pour avoir P=1 bar (P=omeg*ro*eint)  
* 
    flut ro 1.3 eint 0.21978e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 0 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         ros 1630 
         LECT air  TERM 
* 
**  Le TNT :   on donne directement ro = ros 
*              avec ignition au point P7  
*              la vitesse de detonation est celle de Chapman-Jouguet 
* 
    flut ro 1630 eint 3.68e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 0 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         d 6930 TDET 0.0 pini 1e5 xdet 0. ydet 851.0e-3 
         LECT tnt TERM 
* 
LIAISON 
   FSR      LECT  l_cont  TERM      
* 
ECRITURE 
*--   listing : 
       VITE ECROU            TFREQ 200e-6 
            point lect e_capt p8 term 
            elem  lect e_capt term 
* 
*--   dessins : 
       FICHIER ALICE 
         TIME PROG 5E-6 PAS   5E-6   25E-6  
                        PAS  25E-6  200E-6  
                        PAS 100E-6  1.2E-3   
                        PAS 200E-6  2.0E-3   TERM 
* 
*--   courbes : 
      FICHIER ALICE temps 
              tfreq 2e-6 
              point lect e_capt p8 term 
              elem  lect e_capt term 
* 
OPTION NOTEST NOPRINT 
       REZO  mvre   modu  GAM0 0.5  
       DTDROP  0.002 
       AMORT  QUAD 2. 
       LOG 1 
* 
  CALCUL tini 0   pas1 1e-8  nmax 500000   tfin  1.80e-3 

* 
SUITE 
Explosion d'une charge de TNT dans un caisson rigide ( JWLS ) 
* 
ECHO 
* 
RESUL alic temp gardertitre 
* 
SORTIES GRAPHIQUES 
*                                                              
AXTEMPS 1E6 'TEMPS (MicroS)' 
* 
*  Bilan : 
courbe 10 'BILAN '      Bilan             
   TRAC   10                axes 1 'Bilan' 
* 
*  Global : 
courbe 41 'En. int. '   WINT         
courbe 42 'En. cin. '   WCIN      
courbe 43 'En. ext. '   WEXT        
   TRAC   41 42 43          axes 1.e-3 'Global (KJ)' 
*                                                       
*  Pressions (la mesure porte sur les courbes 4 et 5) : 
COURBE  1 'P e_p7  '    ECROU COMP  1   lect e_p7    term 
COURBE  2 'P e_p4  '    ECROU COMP  1   lect e_p4    term 
COURBE  3 'P e_mil '    ECROU COMP  1   lect e_mil   term 
COURBE  4 'P e_p1  '    ECROU COMP  1   lect e_p1    term 
COURBE  5 'P e_p10 '    ECROU COMP  1   lect e_p10   term 
COURBE  6 'P e_face'    ECROU COMP  1   lect e_face  term 
COURBE  7 'P e_p3'      ECROU COMP  1   lect e_p3    term 
COURBE  8 'P e_p12'     ECROU COMP  1   lect e_p12   term 
 TRAC     1 2 3                AXES 1E-5 'P (Bars)'   ! pres du tnt 
 TRAC     4 5 6                AXES 1E-5 'P (Bars)'   ! milieu caisson 
 TRAC     7 8                  AXES 1E-5 'P (Bars)'   ! coins 
*  
*  Densite : 
COURBE 21 'P e_p7  '    ECROU COMP  2   lect e_p7    term 
COURBE 22 'P e_p4  '    ECROU COMP  2   lect e_p4    term 
COURBE 23 'P e_mil '    ECROU COMP  2   lect e_mil   term 
COURBE 24 'P e_p1  '    ECROU COMP  2   lect e_p1    term 
COURBE 25 'P e_p10 '    ECROU COMP  2   lect e_p10   term 
COURBE 26 'P e_face'    ECROU COMP  2   lect e_face  term 
COURBE 27 'P e_p3'      ECROU COMP  2   lect e_p3    term 
COURBE 28 'P e_p12'     ECROU COMP  2   lect e_p12   term 
 TRAC    21 22 23              AXES 1.   'Rho (Kg/m3)' 
 TRAC    24 25 26              AXES 1.   'Rho (Kg/m3)' 
 TRAC    27 28                 AXES 1.   'Rho (Kg/m3)'   
* 
*  Vitesses ; 
COURBE 31 'VY p7   '    VITES COMP  2   lect p7      term 
COURBE 32 'VY p4   '    VITES COMP  2   lect p4      term 
COURBE 33 'VX p8   '    VITES COMP  1   lect p8      term 
COURBE 34 'VY p8   '    VITES COMP  2   lect p8      term 
 TRAC    31 32           AXES 1.   'm/s'     ! faces du tnt 
 TRAC    33 34           AXES 1.   'm/s'     ! coin haut tnt 
* 
* 
VALIDATION 
* 
*  Avec le materiau JWL on ne connait pas de solution analytique 
*  Dans l'experience on mesure les pressions en P1 et P10 
* 
* Pression : 
    ECRO COMP 1  REFE 0.633e5  TOLE 0.20  LECT e_p1   term 
    ECRO COMP 1  REFE 3.092e5  TOLE 0.05  LECT e_p3   term 
    ECRO COMP 1  REFE 0.949e5  TOLE 0.05  LECT e_face term 
    ECRO COMP 1  REFE 0.720e5  TOLE 0.20  LECT e_p10  term 
    ECRO COMP 1  REFE 3.300e5  TOLE 0.05  LECT e_p12  term 
* Masse volumique : 
    ECRO COMP 2  REFE  0.628   TOLE 0.20  LECT e_p1   term 
    ECRO COMP 2  REFE  2.135   TOLE 0.05  LECT e_p3   term 
    ECRO COMP 2  REFE  0.788   TOLE 0.05  LECT e_face term 
    ECRO COMP 2  REFE  0.732   TOLE 0.20  LECT e_p10  term 
    ECRO COMP 2  REFE  2.158   TOLE 0.05  LECT e_p12  term 
* 
FIN 

 

The charge pressures at some selected instants are shown below: 
 

 
Charge pressure at t = 5.E-6 Charge pressure at t = 10.E-6 Charge pressure at t = 20.E-6 

The fluid pressures and velocities at some selected instants are shown below: 
 

 
Pressure at t = 25.E-6 Pressure at t = 50.E-6 Pressure at t = 100.E-6 

 
Pressure at t = 200.E-6 Pressure at t = 800.E-6 Pressure at t = 1800.E-6 

 
Velocity at t = 25.E-6 Velocity at t = 50.E-6 Velocity at t = 100.E-6 

 
Velocity at t = 200.E-6 Velocity at t = 800.E-6 Velocity at t = 1800.E-6 

3D axisymmetric simulation: 
 
JWLS3G 
 
The mesh generation file is: 
 
*%siz 100 
* 
TITRE 'Explosion de TNT dans un caisson' ; 
* 
option echo 0 ; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxvolu3d.proc'; 
option echo 1 ; 
* 
OPTION DIME 3 ELEM CUB8 ; 
* 
OPTION SORTIE 'jwls3g.msh'; 
option trac PSC FTRA 'jwls3d_mail.ps' ; 
* 
p0 = 0 0 0; 
p1 = 0.0355 0 0; 
p2 = 0.965 0 0; 
p3 = 0.965 0 0.709; 
p4 = 0.965 0 0.851; 
p5 = 0.965 0 1.56; 
p6 = 0.0355 0 1.56; 
p7 = 0 0 1.56; 
p8 = 0 0 0.851; 
p9 = 0 0 0.709; 
p10 = 0.0355 0 0.709; 
p11 = 0.0355 0 0.851; 
tol = 1.e-5; 
* 
air1 = dall (p0 d 2 p1) (p1 d 20 p10) (p10 d 2 p9) (p9 d 20 p0) plan; 
air2 = dall (p1 d 26 p2) (p2 d 20 p3) (p3 d 26 p10) (p10 d 20 p1) plan; 
air3 = dall (p10 d 26 p3) (p3 d 8 p4) (p4 d 26 p11) (p11 d 8 p10) plan; 
air4 = dall (p11 d 26 p4) (p4 d 20 p5) (p5 d 26 p6) (p6 d 20 p11) plan; 
air5 = dall (p8 d 2 p11) (p11 d 20 p6) (p6 d 2 p7) (p7 d 20 p8) plan; 
s_air = air1 et air2 et air3 et air4 et air5; 
s_tnt = dall (p9 d 2 p10) (p10 d 8 p11) (p11 d 2 p8) (p8 d 8 p9) plan; 
elim tol (s_air et s_tnt); 
ax_tnt = p9 d 8 p8; 
elim tol (s_tnt et ax_tnt); 
tnt ier1 s1n = pxvolu3d s_tnt 9 90.0 ax_tnt tol; 
tnt = tnt coul roug; 

ax_air1 = p0 d 20 p9; 
ax_air2 = p8 d 20 p7; 
elim tol (s_air et ax_air1 et ax_air2); 
air123 ier2 s2n = pxvolu3d (air1 et air2 et air3) 9 90.0 ax_air1 tol; 
air45 ier3 s3n = pxvolu3d (air4 et air5) 9 90.0 ax_air2 tol; 
air = air123 et air45; 
air = air coul turq; 
elim tol (air et tnt); 
flui = air ET tnt; 
flu8 = flui elem cub8; 
flu6 = flui elem pri6; 
fsrn = enve flui; 
* 
tout = flui ET fsrn ; 
sort tout ; 
* 
trac tnt ; 
trac face cach tnt ; 
trac tout ; 
trac face cach flui ; 
trac QUALIF tout ; 
opti trac mif ftra 'jwls3g_mail.mif'; 
trac tnt ; 
trac face cach tnt ; 
trac tout ; 
trac face cach flui ; 
trac QUALIF tout ; 
* 
mess 'NB_POIN        =' (NBNO tout) ; 
mess 'NB_ELEM (tnt)  =' (NBEL tnt) ;  
mess 'NB_ELEM (air)  =' (NBEL air) ;  
mess 'NB_ELEM (flu8) =' (NBEL flu8) ;  
mess 'NB_ELEM (flu6) =' (NBEL flu6) ;  
* 
**option donn 5 ; 
* 
option echo 1 ; 
* 
fin ; 

 
The input file is: 
 
Explosion d'une charge de TNT dans un caisson rigide ( JWLS ) 
* 
ECHO 
!conv win 
* 
GIBI TOUT 
* 
TRID NONL EULE  
* 
DIMENSION 
    PT3L 13769  zone    2 
    FL38 11664 FL36 432 
    NALE    1  NBLE 13769 
    MTTI 30 
TERM 
* 
GEOMETRIE 
   FL38 flu8 
   FL36 flu6 
TERM 
* 
MATERIAU 
**   l'air :  on calcule eint pour avoir P=1 bar (P=omeg*ro*eint)  
* 
    flut ro 1.3 eint 0.21978e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
         CQ 2.56 PMIN 0 PREF 0 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         ros 1630 
         LECT air TERM 
* 
**  Le TNT :   on donne directement ro = ros 
*              avec ignition au point P7  
*              la vitesse de detonation est celle de Chapman-Jouguet 
* 
    flut ro 1630 eint 3.68e6 gamm 1.35 PB 0 
         ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 

         CQ 2.56 PMIN 0 PREF 0 NUM 11 
         a 3.738e11 b 3.747e9 r1 4.15 r2 0.90  
         d 6930 TDET 0.0 pini 1e5 xdet 0. ydet 0. zdet 851.0e-3  
         LECT tnt TERM 
* 
LINK COUP 
   FSR      LECT  fsrn  TERM      
* 
ECRITURE 
*--   listing : 
       VITE ECROU            TFREQ 200e-6 
            point lect 1 term 
            elem  lect 1 TERM 
* 
*--   dessins : 
       FICHIER ALICE 
         TIME PROG 5E-6 PAS   5E-6   25E-6  
                        PAS  25E-6  200E-6  
                        PAS 100E-6  1.2E-3   
                        PAS 200E-6  2.0E-3   TERM 
* 
*--   courbes : 
      FICHIER ALICE temps 
              tfreq 2e-6 
              point lect 1 term 
              elem  lect 1 term 
* 
OPTION NOTEST NOPRINT 
       DTDROP  0.002 
       AMORT  QUAD 2. 
       LOG 1 
       csta 0.4 
* 
  CALCUL tini 0   pas1 1e-8  nmax 500000   tfin  1.80e-3 
* 
FIN 

The charge pressures at some selected instants are shown below: 
 

 
Charge pressure at t = 5.E-6 

 
Charge pressure at t = 10.E-6

 
Charge pressure at t = 20.E-6 

The fluid pressures and velocities at some selected instants are shown below: 

 
Pressure at t = 25.E-6 

 
Pressure at t = 50.E-6 

 
Pressure at t = 100.E-6 

 
Pressure at t = 200.E-6 

 
Pressure at t = 800.E-6 

 
Pressure at t = 1800.E-6 

 
Velocity at t = 25.E-6 

 
Velocity at t = 50.E-6 

 
Velocity at t = 100.E-6 

 
Velocity at t = 200.E-6 

 
Velocity at t = 800.E-6 

 
Velocity at t = 1800.E-6 

 


